MEASURING CYCLES

By

John Ehlers

INTRODUCTION


Cycles are among the few market parameters that can be quantitatively measured.  That is not to say that their measurement is easy.  In fact, market cycles are so ephemeral that special techniques must be employed to determine that they exist in a sufficiently timely manner to ensure that advantage can be taken of knowledge of the existence.  


In the electronic battlefield, one of the techniques used to avoid detection is to be silent for a long period, transmit information in a short burst, and then go silent again for another long period.  The short transmission time aids in a low probability of detection.  As a countermeasure, special radio receivers have been developed to rapidly set-on and measure new signals before they disappear.  Such devices are called IFM, or Instantaneous Frequency Measurement, receivers.  In this article we will revisit one way of rapidly identifying cycle lengths as well as describing two additional techniques used to build IFM receivers.   EasyLanguage code is presented so you can apply these techniques to your own trading indicators.

A BRIEF REVIEW


One of the fundamental definitions of a cycle is that it has a constant rate change of phase.  For example, a cycle whose phase increases at the rate of 36 degrees per day must be a 10 day cycle because the sum of the day-to-day change of phase completes the full cycle over the 10 day span.  In the first article of this series
 we described a method of obtaining In Phase and Quadrature components of the data using a Hilbert Transform.  The ArcTangent of the ratio of the Quadrature to the In Phase component yields a measured phase angle for each data bar.  Then, by taking the bar to bar difference in phase and summing these differences until 360 degrees is reached gives us an estimate of the dominant cycle period.


Using a schematic diagram shown in Figure 1 is one way of picturing the process we just described.  One can follow the signal flow in the schematic.  The input data stream is first converted to In Phase and Quadrature components and then divided into two paths.  One path is delayed, introducing a phase delay .  The absolute phase angle for the sample is taken in both paths.  When the phase angle of the undelayed path is subtracted from the phase angle from the delayed path, the resultant is the angle .  If the delay is one sample period, then  is exactly the differential phase angle we are seeking.  Therefore, when we sum all the ’s to reach 360 degrees we achieve the dominant cycle estimate by counting the number of times we have to sum the differential phases.  The schematic diagram is a useful way to present the other two IFMs.

I-Q IFM


Since we are describing two cycle measurement techniques, we must give each a name.  I dub this first technique the I-Q IFM because it requires the use of the In Phase and Quadrature components just like the original approach.  It also uses an identical delay.  The schematic diagram for the I-Q IFM is shown in Figure 2.


Recalling, in polar notation, that when two complex numbers are multiplied together their phase angles add.  Since the phase angles are represented as exponents, it follows that phase angles are subtracted when one complex number is divided into another.  The phase angles of the two paths are identical except the delayed path has the additional phase shift .  Therefore, subtracting the phase angle in the undelayed path from the delayed angle in the delayed path exactly yields the differential phase angle  in the output.  All we have to do is then to sum the differential phase angles, as we did originally, to obtain the estimate of the dominant cycle. 


Mathematically, we could do this division several ways.  If we compute the phase angle in each path and then perform the subtraction, we exactly replicate the original approach.  But we want to do something different in the search for a superior approach.  In this case we will literally perform the division.  In the following sequence of equations the original division has both the numerator and denominator multiplied by the complex conjugate of the signal in the undelayed path.  (Complex conjugate means we invert the sign of the imaginary component).  This generates a real number in the denominator.  Since we only care about the argument (i.e. the angle) of the division, we can discard the denominator and only consider the numerator.  Expanding the complex multiplication as shown, the angle  is computed as the ArcTangent of the Imaginary component to the real component.

There are two significant differences between the I-Q IFM and the original approach.  First, the In Phase and Quadrature components in both the delayed and undelayed path are filtered prior to the ArcTangent calculation.  This filtering should reduce noise and yield a more reliable output.  On the other hand, two noisy signals are being multiplied (really, divided), with the result that the multiplication process can introduce additional noise components in the product.


The EasyLanguage code for the I-Q IFM is given in SideBar 1.  We will compare the performance of all the IFMs in a consolidated discussion.

************************************* SideBar 1 ************************************************






EasyLanguage Code for I-Q IFM

Inputs:
Price((H+L)/2);

Vars:
Imult(.635),



Qmult (.338),



InPhase(0),



Quadrature(0),



count(0),



Re(0),



Im(0),



DeltaPhase(0),



InstPeriod(0),



Period(0);

If CurrentBar > 8 then begin


Value1 = Price - Price[7];

 
Inphase = 1.25*(Value1[4]  - Imult*Value1[2]) + Imult*InPhase[3];


Quadrature = Value1[2] - Qmult*Value1 + Qmult*Quadrature[2];


Re = .2*(InPhase*InPhase[1] + Quadrature*Quadrature[1]) + .8*Re[1];


Im = .2*(InPhase*Quadrature[1] - InPhase[1]*Quadrature)   + .8*Im[1];


If Re <> 0 then DeltaPhase = ArcTangent(Im/Re);


{Sum DeltaPhases to reach 360 degrees.  The sum is the instantaneous period.}


InstPeriod = 0;


Value4 = 0;


For count = 0 to 50 begin



Value4 = Value4 + DeltaPhase[count];



If Value4 > 360 and InstPeriod = 0 then begin




InstPeriod = count;



end;


end;


{Resolve Instantaneous Period errors and smooth}


If InstPeriod = 0 then InstPeriod = InstPeriod[1];


Period = .25*InstPeriod + .75*Period[1];


Plot1(Period, "DC");

end;

COSINE IFM


This second implementation of an IFM does not require In Phase and Quadrature components of the input data.  Its performance is described with reference to the schematic diagram of Figure 3.  This description is really an exercise in Trigonometry.  Since the Trig is unavoidable, let’s dive into it.


There are two undelayed paths, one is inverted in phase.  The data in the two undelayed paths are independently summed with the data in the delayed path, resulting in the signals called S2 and S3.  In a radio receiver these signals are detected.  Detection means the signals are squared and then low pass filtered to remove the RF components.  The result is that the detected S3 signal is proportional to Sin2() and the detected S2 signal is proportional to Cos2().  The differential angle is then easily obtained by taking the ArcTangent of the square root of the ratio of the detected signals.  Then, the dominant cycle period is estimated as before by summing the differential phases until 360 degrees is reached.  

When we square these two terms, Cos2(A) = .5*(1 + Cos(2A)).  Note that A has a frequency component of , which is removed by the low pass filter.  As a result of the filtering, only Sin2(/2) remains in the S3 path and only Cos2(/2) remains in the S3 path.  The EasyLanguage code to implement this trigonometry and calculate the dominant cycle from the sum of incremental phases is given in SideBar 2.

************************************SideBar 2***************************************************





        EasyLanguage Code for the Cosine IFM

Inputs:
Price((H+L)/2);

Vars:
Imult(.635),



Qmult (.338),



S2(0),



S3(0),



count(0),



DeltaPhase(0),



InstPeriod(0),



Period(0);

If CurrentBar > 5 then begin


Value1 = Price - Price[7];


S2 = .2*(Value1[1] + Value1)*(Value1[1] + Value1) + .8*S2[1];


S3 = .2*(Value1[1] -  Value1)*(Value1[1] -  Value1) + .8*S3[1];


If S2 <> 0 then Value2 = (SquareRoot(S3 / S2));


If S3 <> 0 then DeltaPhase = 2* ArcTangent(Value2);


{Sum DeltaPhases to reach 360 degrees.  The sum is the instantaneous period.}


InstPeriod = 0;


Value4 = 0;


For count = 0 to 50 begin



Value4 = Value4 + DeltaPhase[count];



If Value4 > 360 and InstPeriod = 0 then begin




InstPeriod = count -1;



end;


end;


{Resolve Instantaneous Period errors and smooth}


If InstPeriod = 0 then InstPeriod = InstPeriod[1];


Period = .25*InstPeriod + .75*Period[1];


Plot1(Period, "DC");

end;

**************************************************************************************************

COMPARITIVE RESULTS


So which cycle measurer is best?  My general decision criteria starts with the premise that “simplest is best”.  Using this criterion, the cycle measurer I originally presented fulfills this requirement.  However, it is possible that some variations of either the I-Q IFM or the Cosine IFM could result in superior performance overall.  Let’s look at how these cycle measurers perform in Figures 4 and 5.  In each case, the Original plots as the red line, the I-Q IFM plots as the yellow line, and the Cosine IFM plots as the cyan line.


The data in Figure 4 is a theoretical sinewave that switches back and forth between a period of 15 bars and a period of 30 bars.  All three IFMs have nearly identical performance in this test.  It should be noted that all three approaches correctly identify the period of the theoretical cycles.


The signal in Figure 5 is a real-world example of data that includes both Trend Mode and Cycle Mode conditions.  Now there is a decided difference in performance, and it shows just how difficult cycle measurement is.  While it is difficult to ascertain which one is most correct, I have observed that the I-Q IFM and Cosine IFM are bias toward the longer cycle lengths.  When I increase the amount of smoothing in these IFMs the dominant cycle period is reduced in some cases.  Of course, this causes an additional lag in the measured cycle period.   From this, I conclude that all data contains both signal and noise, expressed as (S+N).  When we take cross products of two data streams in our calculations, the signal plus noise components are of the form S2+N2+2SN.  Now, the noise is intermixed with the signal and better filtering is required.  Perhaps there is a better filtering approach available, and I hope that maybe you can identify it.  Nonetheless, the process of eliminating noise is one of filtering bandwidth.  Reducing the effective noise bandwidth is exactly what the MESA algorithm does while performing its spectral estimate.  Therefore, the IFM approaches described in this article can be used for “quick and dirty” estimates of the dominant cycle, but MESA must be used if greater precision is required.

Figure 1.  Schematic Diagram of a method to measure differential phase.

Figure 2.  Schematic Diagram if the I-Q IFM

Figure 3.  Schematic Diagram of the Cosine IFM

Figure 4.  IFM Measurements of a Theoretical Sinewave having an abrupt change in Dominant Cycle

Figure 5.  Real World Noise disrupts IFM Measured Performance

� John Ehlers, “Signal Analysis Concepts”

